Exercises

To derive the maximum benefit from these exercises, work through them sequentially. This is especially true for new users who need time and practice to become accustomed to Maple syntax and the idiosyncrasies of the Maple interface.

Solutions to the Exercises marked with an asterisk will be found in the Solutions section.

An important reminder for PC users: Any reference to the **Command** key should be replaced with the **Control** key.

Part I. The Maple Worksheet

Section 1. Execution Groups: Input/Output

0. Pull down the **Help** menu and choose **Using the Help System** on the **Manuals, Dictionary, and more** submenu.

Open a new Maple worksheet, open the **Preferences...** dialogue, choose **Display** and set the *input* display to **Maple Notation** (as opposed to **2-D Math Notation**). Click on **Apply Globally** at the bottom of the panel. Then dismiss the dialogue and return to the worksheet to do the following.

1. Reduce the fraction 546/1001 to lowest terms by typing

```
546/1001;
```

at the input prompt and pressing [return]. Press [Enter] on a PC. This is referred to as executing the entry.

2.* Simplify the square root of 19,220 by entering and executing

```
sqrt(19220);
```

Then obtain the complete integer factorization of 19220 by entering and executing

```
ifactor(19220);
```

Read **ifactor** as "integer factors of".

3.* Obtain an integral of the expression

$$e^{3x} \cos(2x) \sin(4x) .$$

Enter it into the **int** procedure as follows

```
int( exp(3*x)*cos(2*x)*sin(4*x), x);
```

4. Pull down the **View** menu. Choose **Hide/Show Contents** and use the mouse to *deselect* **Section Boundaries** and **Execution Group Boundaries**. Dismiss the dialogue and make note of the changes in the appearance of the worksheet.

5. Pull down the **View** menu once more. Choose **Hide/Show Contents** again and *deselect* **Input**. Make note of the changes in the appearance of the worksheet.

6. Restore the worksheet to its original form (show **Section Boundaries**, **Execution Group Boundaries**, and **Input**). Save it with the name "My_1st_Mpl_wrksht"

Section 2. Entering blocks of text

1. Open a new Maple worksheet. Set the input display to **Maple Notation** as in Section 1, Exercise 1. Then press **Command-T** and type your name, press the [return] key and type the date. Press [return] and type "Mixing Text and Mathematics" (no quotes). Then press the [return] key one more time and type the following sentence:

> This worksheet contains examples of text blocks (like this one) and execution groups containing input and output like the following.

2. Now press **Command-J** to obtain an input entry and enter the variable y as the following function of x:

$$y = \frac{\sin(x)}{x}\ .$$

Note that you will have to type

```
y := sin(x)/x;
```

then press the [return] key ([Enter] on a PC). We will refer to this as "enter and execute".

3. Before continuing use the mouse to click anywhere in the third text line at the top of the worksheet, "Mixing Text and ....". Then pull down the menu of paragraph and text styles on the left side of the Context bar and choose the paragraph style named **Title**.

4. Go back down the worksheet and click next to the new input prompt that appeared after the entry defining y as a function of x and define z as the derivative of y with respect to x by entering and executing

```
z := diff(y,x);
```

5. When the new execution group appears below the derivative formula for z, press **Command-T** to convert it to a text entry. Then type the following sentences:

> z is the derivative of y. Both z and y are plotted below. Which one is which? How can you tell?

6. Press **Command-J** and graph y and z as functions of x by executing the entry

```
plot( [y,z], x=0..6);
```

7. Press **Command-T** once more (to convert the new execution group into a text entry) and type in your answers to the two questions asked above.

8. Save this worksheet with the title "My_2dn_wrksht".

9. Quit Maple.

10. Open the worksheet you just saved by double clicking on its icon.

12. Pull down the **Edit** menu and choose **Remove Output/From Workshee**t. (At the very bottom of the menu.)

13. Pull down the **Edit** menu again and choose **Execute/Worksheet**. This puts the information in the worksheet back into the Maple kernel.

14. There should be a new execution group immediately following the graph. If the cursor is not at the input prompt, use the mouse to click next to the prompt and press **Command-delete**

thereby removing that empty execution group from the worksheet.

15. Now, with the mouse, click anywhere in the last text entry, press **Command-J** and, in the new execution group, define the variable Y as an antiderivative of z with the following entry

Y := int(z,x);

16. Press **Command-T** and type the following: Well, that wasn't hard. Now I now how to enter text and mathematics in Maple.

17. Pull down the **File** menu and if you are on a PC choose **Print Preview...** . If you are on a Mac choose **Print...** and click on **Preview** in the ensuing **Print Dialogue**. When you see the print preview you will discover that the graph is very large causing the worksheet to be two pages long when printed. Dismiss the preview and make the graph small enough so the document prints in just one page. (Click on the graph and adjust its size by dragging the lower right corner.)

18.* Save your worksheet (**Command-S**) then print it.

Part II. Calculations and Calculus with Maple

Section 1. Getting Started: Maple as a Calculator

The following exercises provide practice using Maple to make simple calculations like the ones in the manual. Compare the output to what you can get from your calculator.

Open a new Maple worksheet (**Maple Notation** for the input...**Preferences/Display**) and to the following.

1. Enter the following sequence of square roots.

$$\sqrt{24}, \sqrt{864}, \sqrt{555}$$

Then enter and execute

```
evalf(%);
```

Now execute

```
evalf[5](%%);
```

Now execute

```
add(k,k=Label);
```

where **Label** is the Label assigned to the first input: (1). Note that you cannot enter this by simply typing the symbols `(1)`. You must use the **Insert** menu (or press **Command-L**) and type the number in the Label dialogue.

Read the Help page for **add** and comment on what Maple did to get the output to the **add** input.

2. Execute the following entry

```
x = 4/(1+sqrt(2)) + exp(3) - ln(4);
```

Execute

```
evalf[4](%);
```

Now execute

```
x;
```

and note that x is still a free variable. This illustrates the fact that

Equations do not assign values.

3. Make the same entry that starts exercise 2. Then execute

simplify(%);

Is the output any simpler, in your opinion?

4. Make the same entry that starts Exercise 2. Then execute the following two entries. They can be entered at the same input prompt.

```
rationalize(%); expand(%);
```

Comment on what Maple did to get the two outputs. (Remember what "rationalize the denominator" means from high school algebra.)

5. Execute the following entry.

```
(1 + sqrt(3))/(1 - sqrt(3));
```

Then execute

```
rationalize(%);  expand(%);
```

6. Use the sequence operator, `$`, to make the following sequences.

a. 2, 4, 6, 8, 10, 12

b. 20, 40, 60, ... , 260   Hint. 20 + 20*k $ k=

c. Now use the sequence procedure, **seq**, to make the sequence in part b.

7.* Add the numbers in the sequence in 6 c. Hint. If you just did 6 c, then execute **add(k,k=%)**.

8. Multiply the numbers in the sequence in 6 c. Then execute

length(%);

to find out how many digits there are in the product.

9. Use the sequence procedure, **seq**, to obtain the sequence in Exercise 6 a.

10. Use **seq** to obtain the sequence of the cubes of the numbers in the list [2, 5, 6, 9, 12, 44]. Add the numbers in the sequence of cubes and then obtain the prime factorization of the sum using **ifactor**.

What do you notice about the prime factorization? (Hint. The integer 87990 is called "square free".)

11.* Obtain the prime factorization of the *product* of the integers in the sequence of cubes described in Exercise 10.

12. Use the **factor** procedure to factor the following polynomial expressions.

a. x^3 - x^2 + x - 1

b. x^7 - x^6 + x^5 - ... -1   Hint. Enter this as **`add( -(-x)^k,k=0..7)`** then **`factor(%)`**.

13. Use **factor/real** to factor the two polynomials in Exercise 12 over the real field.

14. Use the **factor/complex** to factor the two polynomials in Exercise 12 over the complex field.

15.* Use **solve** to obtain the zeros of the polynomials in Exercise 12.

16.* Add the zeros of the polynomial in 12 a and the zeros of the polynomial in 12 b.

17. Multiply the zeros of the polynomial in 12 a and the zeros of the polynomial in 12 b.

Section 2. Symbolics: Equations and Assignments

Solving equations is the bread and butter of mathematics. Maple does it in a natural way. It is always a good idea to assign a name to the equation and the solution.

Open a new Maple worksheet and do the following.

1. Make the following entry.

```
restart; x := y - z;  y = 3: z := 4;
```

Then enter and execute x and explain the output.

2. Make the following entry.

```
restart; x := y - z;  y := 3: z = 4;
```

Then enter and execute x and explain the output.

3. Make the following entry

```
restart; x := y - z;  y := 3: z := 4;
```

Then enter and execute x and explain the output.

Now enter the following

```
restart;  x := z:  z := y - w:  w := 5:
```

Write on a pad of paper what you think Maple has stored for x, y, z, and w. Then predict the output for the entry

```
x + z;
```

Execute this entry to check your guess. Then do a restart by clicking on the *restart button* on the button bar (just to the right of the bug).

4. Enter the equation

$$x^3 - x^2 + x - 1 = 0$$

with the name *eqn*.

Solve the equation and name the solutions solns with the entry

```
solns := solve( eqn, {x});
```

Check the third solution with the entry

```
eval( eqn, solns[3]);
```

Explain why the output confirms the validity of the solution.

Check all three solutions with the entry

```
eval(eqn,solns[k]) $ k=1..3;
```

Read the error message carefully and fix the problem (*premature evaluation*) by enclosing the **eval** procedure in single quotes to delay evaluation as shown below . Do not retype the entry, just add the single quotes and press the [return] key.

```
'eval(eqn,solns[k])' $ k=1..3;
```

5. Use **fsolve** to obtain an approximate real solution to the equation

$$x^3 - 0.9\,x^2 + x - 1 = 0$$

Name the solution *soln*.

Hint. Name the equation *eqn* and simply enter **soln := fsolve(eqn,{x})**.

Check the approximate solution using `eval(eqn,soln)`.

6.* Plot the expression

$$x^3 - x^2 + 0.05\,x + \cos(x) - 0.7$$

with the entry

```
y := x^3 - x^2 + 0.05*x + cos(x) - 0.7; plot(y,x=-2..2,-2..2);
```

Enter and execute `fsolve(y=0,x)` to see which zero **fsolve** finds.

Using the graph as a guide, obtain an approximation to the largest positive zero using

```
fsolve( y=0, {x=a..b} )
```

7. Find the first three positive solutions to the equation

$$\cos(x) = x\tan(x)$$

Hint. Define the function $y = \cos(x) - x\tan(x)$ with the entry

```
y := cos(x) - x*tan(x);
```

Plot *y* using appropriate domain and range values (and `discont=true`), then use **fsolve** with specific interval settings obtained from the graph.

8. Let y be the function of x defined in Exercise 7. Enter

```
solve(y=0,{x});  evalf(%);
```

and comment on the output.

9. Graph the function

$$y = \cos(x^2) - \sin(2\,x)$$

over the interval from x = 0 to x = 2. Define *yp* as its derivative and find the zero of *yp* near x = 1.5. Name it *xmin*. Then find the minimum value of y over this interval using `eval(y,x=xmin)`. Name it *ymin*.

Use the following entry to plot the graph and the low point.

```
plot( [y, [ [xmin,ymin] ]  ], x=0..2, style=[line,point]);
```

10.* Find the area of the region between the graph in Exercise 9 and the x-axis.

Hint: Numerically integrate the absolute value of y from x = 0 to x = 2 via the entry

```
evalf( Int( abs(y), x=0..2) );
```

Section 3. Functions as Transformations

Functions play a key role in many applications of mathematics. The arrow notation is used to make functions and the unapply procedure can turn any expression into a function. The Matrix procedure can be used to make tables of data.

Open a new Maple worksheet and do the following

1. Use arrow notation to define the function $f(x) = cos(x) - x \; tan(x)$. Then use the entry

```
g := D(f);
```

to obtain g as the derivative function.

2. Continuing 1. Plot f over the interval x = -1 to x = 1 using **`plot( f(x), x=-1..1)`**.

Then find the area of the region below the graph of f and above the x axis.

Hint. This will require the values b where f(b) = 0. Find the positive value using

```
b := fsolve(f(x)=0,x,0..1);
```

Check that $f(-b) = -1$ also. Once b is found, the area is the integral of f from $-b$ to b.

3. Continuing 2. Plot the graph of f from -1 to 1 and the tangent line segment to the graph at the point x = 0.5, y = f(0.5) over the interval x = 0 to x = 1.

Hint. You already have the derivative function g. Use it to define the tangent line function using

```
T := x -> f(0.5) + g(0.5)*(x - 0.5);
```

Then execute

```
plot( [f(x), [(t,T(t),t=0..1] ], x=-1..1);
```

You will see that Maple's default colors for two curves are red and green. The second curve is easier to see with the optional equation **`color=[red,blue]`**. Try it, you will like it.

4. Continuing 3. Find the length of the curve plotted in Exercise 2.

Hint. Do the integration numerically using **`evalf( Int( .... ) )`**.

5.* An animation.

The tangent line plot in Exercise 3 can easily be animated as follows.

First define the function $T(a, x)$ whose value at (a, x) is the formula for the tangent line to the graph of f at the point $(a, f(a))$.

```
T := (a,x) -> f(a) + g(a)*(x - a);
```

Then use

```
plots[animate]( plot, [ [f(x), [t,T(a,t),t=a-0.5..a+0.5] ],
                          x=-1..1,-1..2,
                          color=[red,blue] ],
                 a = -1..1);
```

Once the plot appears, click on it with the mouse to change the context bar into a row of video controls. Experiment, enjoy.

Read the Help page for **plots[animate]**.

6. Solving a differential equation.

The **dsolve** procedure solves differential equations. The syntax is simply

```
dsolve( deqn )
```

where *deqn* is a differential equation or the name of one. Do a restart and define a simple first order differential equation as follows

```
restart;  deqn := diff(y(x),x) + x*y(x) = x;
```

Get the general solution with the entry

```
dsolve( deqn );
```

Now obtain the solution satisfying y(0) = 0 using

```
soln := dsolve( {deqn, y(0)=0} );
```

7. Plot the last solution using

```
plot( rhs(soln), x=-2..2);
```

8.* Using **unapply** and **Matrix**.

Convert the second solution in Exercise 6 into a function f using

```
f := unapply(rhs(soln),x);
```

Evaluate the solution at x = 0, 0.2, 0.4, ... , 1.0 to 4 digits as follows

```
evalf[4](  f(0.2*k) $ k=0..5 );
```

Make a table of function values with the following Matrix entry

```
Matrix( [ [x, 0.2*k $ k=0..5] , [ 'f(x)', % ] ] );
```

Part III. First Order Ordinary Differential Equations

Section 1. Entering, Solving, Plotting

Unevaluated derivatives are used to enter a differential equation, **dsolve** solves it. The phi function can be used to plot families of solutions satisfying specified initial conditions.

Open a new Maple worksheet and do the following

1. Enter the differential equation $y' + y = sin(x)$ using the **D** operator. Name it *DE1*. Obtain the general solution and the solution satisfying $y(0) = 0$.

2. Continuing 1. Enter the same differential equation using the **diff** procedure. Name it *DE2*. Obtain the general solution and the solution satisfying $y(0) = 0$. Compare the solutions to those found in Exercise 1.

We recommend that you enter differential equations using the **diff** procedure.

3. Obtain the general solution to the equation $y' = \frac{x}{cos(y)}$ by entering the equation with the name DE and using

```
dsolve( DE );
```

Now solve the equation using

```
dsolve( DE, implicit );
```

4. Continuing 3. Obtain the solution to DE satisfying y(0) = 1. Do it twice, once using

```
dsolve( {DE,y(0)=1} );
```

and again using

```
dsolve( {DE,y(0)=1}, implicit );
```

5. Obtain the general solution to the equation

$$\frac{d}{dx} y(x) = \frac{y(x)^2 + 1}{x^2 + 1}$$

and plot the solutions corresponding to _*C1* = -2, -1, 0, 1, 2.

Hint. Experiment with the horizontal plot range until you get nice picture displaying all 5 curves near x = 0. Use the optional equation **axes=framed**.

6. Continuing 5. Obtain the solution to the differential equation satisfying *y*(1) = 1. Plot it over a reasonable interval containing x = 1.

Use **unapply** to convert the solution into a function f. Use f to generate a sequence of solution values for x= 0, 0.25, 0.5, 1.25, 1.50, 2.0.

7.* Enter the differential equation $y' + y = cos(x)$ with the name *DE*. Obtain the solution satisfying the generic initial condition $y(x0) = y0$. Name it *soln*.

a. Use the entry `eval( soln, x=x0)` to verify that *soln* satisfies the initial condition.

b. Use **unapply** to make `rhs(soln)` into a function of x, x0, and y0 named phi (see pages 36-37 in Part III, Section 1 of the manual).

c. Use phi to plot some solutions starting at points evenly spaced on the y axis.

d. Use phi to plot some solutions starting at points evenly spaced on the x axis.

e. Use phi to plot the solutions starting at points evenly spaced around the unit circle in the style of the two plots on page 37 of the manual. That is, one picture runs x forward, another runs x backward.

8. Continuing 7. Examine the formula for the phi function and describe the behavior of all solutions as x approaches infinity.

9. Look before you leap.

Consider the following differential equation.

$$\frac{d}{dx} y(x) = \frac{y(x)}{x - 2}$$

a. What is the formula for the function *f* such that this equation is equivalent to

$$\frac{d}{dx} y(x) = f(x, y(x)) \text{ ?}$$

b. Based upon the statement of the Unique Solution Theorem at what points do you expect that solutions will fail to exist and/or fail to be unique?

c. Obtain the phi function for this differential equation and use it to obtain the solutions described in parts c, d, and e of Exercise 7.

10. Obtain an informative picture of the family of solutions to

$$\frac{d}{dx} y(x) = \frac{y(x)}{x^2 - 1}$$

over the interval $x < -1$. Hint. Use the phi function. (See page 37 in the manual.)

Section 2. Working with Solutions: Modeling

Getting the solution formula is only the beginning of the story in most applications. Mathematical modeling requires solution manipulation. The exercises in this section are similar to the examples in the manual.

Open a new Maple worksheet and do the following

1. Joe's savings account contains $12,000 dollars. He does not know the interest rate (compounded continuously) but after 60 days he checks and discovers that he has $12,130.

   a. What is his annual interest rate?
   b. How much money will be in the account one year from the day he has $12,130?

   c. Plot the graph of P(t) including the points corresponding to 60 days and 60 days + 1 year.

Hint. First solve the exponential model initial value problem $P' = r\,P$, $P(0) = 12000$ with variable r and then substitute the data for $t = 60/365$ to determine the value of r. If you call the solution "soln" the substitution and r calculation can be made as follows

```
subs(P(t)=12130, t=60/365, soln);
r := evalf( solve(%,r) );
```

Then calculate $P(1+60/365))$ using

```
soln := subs(%,soln);  eval( soln, t=1+60/365);
```

The graph is the plot of **`rhs(soln)`** (and the two points).

2. Continuing 1. In Exercise 1 you discovered that Joe's savings account has an annual interest rate of 6.555 percent (approximately). Suppose that on the 60th day Joe also has a credit card debt of $560 at a 9% annual interest rate. Starting on that date he decides to pay off the credit card debt continuously from his savings account at the annual rate of $400 per year.

   a. How many days later will the credit card debt be paid off?

   b. At the time the credit card debt is paid off, how much money will Joe have in his savings account?

3.* Suppose you borrow $12,000 to buy a car. The loan is to be paid in 60 equal monthly installments at an interest rate of 5% per year.

   a. Assume the payments are actually made continuously at whatever rate is needed to pay off the loan in 60 months. Determine the continuous rate per month that would be required.

   b. Compare the answer to part a to the answer if 60 equal monthly payments are made at a constant annual interest rate of 5% applied to the outstanding balance. In other words, the first payment, due one month after the loan is made, would be

$$12000\left(e^{\frac{0.05}{12}} - 1\right) + P_1$$

   where P[1] is the amount that is put towards reducing the principal (the $12000) in the first month. The second payment is

$$\left(12000 - P_1\right)\left(e^{\frac{0.05}{12}} - 1\right) + P_2$$

where P[2] is the amount put towards reducing the principal in the second month. The nth payment is

$$\left(12000 - \left(\sum_{k=1}^{n-1} P_k\right)\right)\left(e^{\frac{0.05}{12}} - 1\right) + P_n.$$

Hint for b. The second payment is supposed to equal to the first. Therefore, the following equation must be satisfied.

```
> eqn := 12000*(exp(0.05/12)-1)+P[1] =
        (12000 - P[1])*(exp(0.05/12)-1) + P[2];
```

$$eqn := 50.10431 + P_1 = 50.10430800 - 0.004175359\, P_1 + P_2 \quad (3.2.1)$$

This determines P[2] in terms of P[1].

```
> ?finance
> P[2] := solve(eqn,P[2]);
```

$$P_2 := 0.000002000000000 + 1.004175359\, P_1 \quad (3.2.2)$$

Create a **for..do** loop that calculates P[3], P[4], ... , P[60] in terms of P[1], then find P[1] using the fact that

$$\sum_{k=1}^{60} P_k = 12000$$

Solution. The answer to part a. is \$226.04 per month. The answer to part b. is \$226.51 per month or about \$30 more over the five years of the loan. See the Solutions.

4. Continuing 3. Consult with a banker and/or the internet to determine the amount a bank would charge per month for a 5% loan of \$12000 over 60 months. Maple has a package called **finance**. Execute the entry

```
?finance
```

and read the Help page for this package. Can any of the package procedures be used here?

5. There is a power failure in your house at 1:00 P.M on a winter afternoon and your heating system stops working. The temperature in your house is 68 degrees F when the power goes out. At 10:00 P.M. the temperature in the house is down to 57 degrees F. Assume that the outside temperature is 10 degrees F.

a. Estimate the temperature in your house at 7:00 A.M. the next morning. Should you worry about your water pipes freezing?

b. Suppose the power goes on at 8:00 A.M. providing heat flow into the house that would increase the temperature at the rate of 10 degrees per hour if there were no heat loss. At what time will the temperature in the house be 68 degrees F again assuming the outside temperature stays at 10 degrees F throughout the day?

c. What is the answer to the question in part b if the heating system provides heat sinusoidally with a period of 3 hours and a maximum temperature increase of 10 degrees:

$$H(t) = 10 - 10\cos\left(\frac{2\pi t}{3}\right)$$

6. Verify that there is only one value of k that satisfies the *k_equation* on page 47 of the manual.

Hint. Plot, on the same set of axes, the left side and the right side of *k_equation*.

Section 3. Slope Fields: DEplot

The **DEplot** procedure draws slope fields and solution curves directly from the ode.

0. Enter **`?DEplot`** and read the Help page.

Open a new Maple worksheet and do the following.

1. Consider the following first order differential equation.

$$(x+1)\left(\frac{d}{dx}y(x)\right) = 1 - y(x)$$

a. Use **dsolve** to obtain the general solution and also the solution satisfying y(0) = 0. Plot the second solution using **`x = -2..2, -2..2`** and compare the picture to the solution curve shown below.



b. Attempt to reproduce the picture above using **DEplot**.

c. Get a better picture by doing the following. Draw the six solution curves in one plot using the phi function for the differential equation. Immediately after making a plot that you like, use a **DEplot** entry to make the slope field (**`arrows = line`** and no solution curves), but *terminate the entry with a colon.* Then make the following entry

```
plots[display]( %, %% );
```

d. Comment on the existence and uniqueness of solutions using the statement of the Unique Solution Theorem. Your comments should be based upon the properties of the function $f(x, y)$ where $y' = f(x, y)$.

2.* Use **DEplot** to make a nice-looking slope field for the autonomous equation $y' = sin(y)$. (Use the window **`x=-6..6, y=-6..6`**.)

a. Put some solution curves into the plot by adding

```
{ [y(0)=k] $ k=-6..6 }, linecolor=blue.
```

Comment on the relationship between one curve and the next.

b. Obtain the general solution using **dsolve**.

c. Use **dsolve** to obtain the solution satisfying the initial condition y(0) = 1. What is the value of this solution when x = 1? Get the exact value and an approximation. If the solution is called "soln" use

```
eval(soln,x=1); evalf(%);
```

d. Comment on the long-term behavior of solutions to this differential equation.

3. Repeat Exercise 3 using the differential equation $y' = 1/\sin(y)$. Use the same plot window and inits, but reduce the stepsize to 0.1. Then make two more **DEplots** in the window **`x=0..3, y=0..4`** with only two inits, y(0)=1 and y(0)=2 and stepsize 0.1. First use the default numerical method, then use the optional Runge-Kutta Fehlberg method by adding the equation **`method=rkf45`**. (Copy and Paste to make the new entry.)

Note that the symbolic solutions are much simpler than the ones in Exercise 3. Explain why.

Section 4. Approximate Solutions

Looping constructs and user defined procedures are featured in this set of exercises.

Open a new Maple worksheet and do the following.

1. Create the procedure named *Euler* defined on in Part 3, Section 4 of the manual. Test it on the differential equation $y' = \cos(x)y$ with the initial condition $y(0) = 1$ (as in the manual).

2. Use *Euler* to obtain the list *L1* of Euler approximations to the initial value problem (IVP)

$$\frac{d}{dx}y(x) = \frac{8\,e^{-x}}{3+y}\ ,\ y(0)=1\ .$$

for $x = 0, 0.2, 0.4, 0.6, 0.8, 1.0$. (I.e. $h = 0.2$). Plot these points and the line segments connecting them along with the actual solution. Name the plot *P1* by entering the following

```
    plot( [L1,rhs(soln),L1], x=0..1, y=0..2, style=[line$2,
point],
           color=[red,blue,black]);  P1 := %:
```

Be sure to terminate the definition of *P1* with a colon to suppress the output.

Note. You will have to create the solution using **dsolve**.

3. a. Make a similar plot named *P2*, displaying the solution and the approximation for $h = 0.1$. (Call the list of approximations *L2* to distinguish it from the list in Exercise 2.)

b. Use the **display** procedure to display plots *P1* and *P2* together. Enter the following

```
plots[display]( P1, P2 );
```

Comment on the error displayed in the picture.

4. a. (Continuing 3) Use **unapply** to make **`rhs(soln)`** into a function g. Then display the entries

in list *L1* and half of the entries in *L2* (the ones corresponding to the same *x* values) as well as the actual solution values in a matrix as follows:

```
Matrix( [ [  t , 0.2*k $ k=0..5 ],
          ['h=0.2', 'L1[k,2]' $ k=1..6 ],
          ['h=0.1', 'L2[2*k+1,2]'$ k=0..5 ],
          ['g(x)' , g(0.2*k) $ k=0..5 ] ] ): evalf[4](%);
```

b. Use your calculator to check that the error has been cut in half (approximately).

5. Create the procedure called *ImpEuler* and test it on $y' = cos(x)\ y$ with the initial condition $y(0) = 1$ (as in the manual).

6. Repeat 2 - 5 using *ImpEuler* in place of Euler. Comment on the graphs and the reduction of error displayed in the matrix.

Hint. Use **Copy** and **Paste**, then make minor changes in the code.

7.* Modify the *ImpEuler* procedure to make a procedure that implements the classical Runge-Kutta algorithm (see Simmons/Krantz, Chapter 9, Section 5). Test in by applying it to the problems described in Exercises 2 - 6. (Copy and Paste).

▼ Part IV. Linear Differential Equations

▼ Section 1. Linear Oscillators

The harmonic oscillator is the fundamental model for the analysis of oscillating systems. Phase plane trajectories are constructed. **DEplot** draws direction fields.

1. Obtain the solution to the following initial value problem. Call it *soln*.

$$y'' + 4y = 0 \ \ , y(0) = 2 \ , y'(0) = -3$$

From the form of the solution decide if the system is undamped, underdamped, critically damped, or overdamped. What is the period of the oscillations?

Continuing 1. Plot the solution to the IVP in Exercise 1. Then create a plot showing the cosine term, the sine term and their sum (the solution curve). Make the solution red, the cosine blue, and the sine green.

a. What IVP does the cosine term solve?

b. What IVP does the sine term solve?

c. Determine the amplitude of the oscillations by solving $y'(t) = 0$ and substituting the time value into the solution. Compare the answer to the amplitude calculated using the standard formula for converting the solution into amplitude/phase angle form.

d. Assuming this is the model of a mass spring system, determine the speed of the mass as it passes through equilibrium.

e. Use **unapply** to convert the solution into the function g. Use g to plot the phase plane trajectory. What type of curve is this trajectory?

f. Add to the trajectory the points corresponding to $t = 0, 0.25, 0.5, 0.75, \ldots, 2.0$.

2.* Consider now the following damped system. Obtain the solution.

$$y'' + y' + 4y = 0 \;\; , y(0) = 2 \; , y'(0) = -3$$

From the form of the solution decide if the system is underdamped, critically damped, or overdamped.

a. What is the pseudo-period of the oscillations?

b. What is the time constant?

c. Based upon your answer to part b estimate the time interval required for the oscillations to disappear from view.

d. Plot the solution curve over the interval you named in part c.

e. Add to the curve in part d the curves defined by $A \; e^{-\frac{t}{2}}$ and $A \; e^{\frac{t}{2}}$ where

$$A = \sqrt{4 + \frac{16}{15}} \, .$$

Make them blue. What is the significance of these curves? Where did the formula for A come from?

f. Use **unapply** to convert the solution into the function g. Use g to plot the phase plane trajectory.

g. Add to the trajectory the points corresponding to $t = 0, 0.25, 0.5, 0.75, \ldots , 2.0$.

3. Use **DEplot** to draw the direction field in phase space for the undamped system. Then add the solution trajectory.

Note. You will have to enter the equivalent system of two first order equations. See the manual, pages 76-77.

4.* Use **DEplot** to draw the direction field in phase space for the damped system. Then add the solution trajectory.

Section 2. State Space

The forced oscillator is modeled with a non-autonomous equation. Solution trajectories are best viewed in state space.

1. Consider the driven IVP

$$y'' + 4y = \cos(1.8\,t) \;\; , y(0) = 2 \; , y'(0) = -3$$

Obtain the solution, call it *soln* and convert it into a function g using **unapply**.

a. Plot the solution curve over the interval t = 0..120. What you witness in the plot is the phenomenon called "beats". The output pulsates like this when the driver frequency is very close to the natural frequency of the system and the system is lightly damped.

b. Obtain the phase plane trajectory for this system. Use the same time interval.

c. Obtain the state space trajectory for this system. Use the same time interval. Add the equations

```
numpoints=800, axes=framed, orientation=[-60,70],
labels=["Position","Velocity","Time"].
```

2.* Damp the system slightly by changing the IVP to the following

$$y'' + 0.1y' + 4y = \cos(1.8\,t) \quad , y(0) = 2 \; , y'(0) = -3$$

Obtain the solution, call it soln and convert it into a function g using unapply. Suppress the output for soln and the definition of g and then enter

```
evalf[3]( 'g(t)' = g(t) )
```

to see a nice looking representation of the solution formula.

a. Based upon the solution formula determine the time constant for the beats. That is, how long will it take (approximately) for the beats to disappear from the solution curve as it settles down to its steady-state mode?

b. Plot the solution curve to verify your answer to part a.

c. Use g to obtain the phase plane and state space trajectories for this system.

3. Define the procedure called *DEsystem* in Part 4 Section 2 of the manual. Apply it to the differential equation in Exercise 1 of Section 1 (this Part) and use **DEplot** to obtain the direction field and the solution trajectory. (The direction field can be drawn in phase space because the equation is autonomous.)

4. Apply *DEsystem* it to the differential equation in Exercise 1 in this section and then use **DEplot** to obtain the solution trajectory. (The direction field cannot be drawn in phase space because the equation is not autonomous.)

5.* Apply *DEsystem* it to the differential equation in Exercise 2 in this section and use **DEplot** to obtain the solution trajectory.

6.* Continuing 5. Apply **DEplot3d** to obtain the state space trajectory for the IVP in Exercise 2.

7. Use **dsolve/numeric** to obtain a numeric solution for the IVP in Exercise 2. Apply **odeplot** to sketch the time series for position and the time series for velocity. Then obtain the phase plane trajectory and the state space trajectory using **odeplot**.

8. Obtain a nice display of numerical solutions to the IVP in Exercise 2 by using **dsolve/numeric** with the

```
output=array( list )
```

option. Make the display even nicer by suppressing the output and then entering

```
evalf[4](%)
```

Section 3. Two Dimensional Systems

A two dimensional system defines a two dimensional vector field and a vector flow. The **dsolve/numeric** procedure outputs approximate solution values and curves. Linear systems can be solved using standard methods by reduction to one second order equation or by matrix methods (featured in the next section).

Load the **plots** and **DEtools** packages via **`with(plots): with(DEtools):`**

1. Use **DEplot** to obtain the phase portrait for the linear system

$$x' = -x + 2y$$
$$y' = 4x + y$$

Note. The phase portrait is a sketch of some solution curves and the direction field when the system is autonomous. Compare the picture you get to our version reproduced below.



2. Continuing 1. Use **dsolve** to obtain the general solution to the system in Exercise 1. Call the solution *soln* and represent it as a column vector using the entry

```
subs(soln, <x(t),y(t)> );
```

3. Add the two nullclines to the phase portrait in Exercise 1. (What is the stationary point?)

Hint. Go back to the picture of the flow and add the entry **`PP := %:`** at the same input prompt and directly after the **DEplot** entry creating the phase portrait. Execute the entries. Then make the nullclines using **implicitplot** (both can be drawn at the same time). Call the nullcline plot *NP* and then use

```
display( PP, NP );
```

4.* The model

$$x' = x\left(1 - y - \frac{x}{a}\right)$$
$$y' = y\left(1 - x - \frac{y}{b}\right)$$

is used to study competing species. The constants a and b are positive. Use **DEplot** to do the following.

a. Draw the direction field when $1/a = 1.9$ and $1/b = 1.5$. Use **`x = 0..1 , y = 0..1`**.

b. Find the stationary point and add the nullclines to the direction field. Discuss the solutions based upon the picture you see.

c. Add solution curves corresponding to the following set of initial conditions (a circle of points around the stationary point)

```
{[0,0.3+0.2*cos(Pi/6*k),0.5+0.2*sin(Pi*k/6)]$k=0..11}
```

5. Use **DEplot** to obtain the phase portrait for the system

$$x' = x - 2y - 1$$
$$y' = x - y - 2$$

Add the stationary point and nullclines. Find the solution formulas using **dsolve**.

This is a linear system with periodic solutions (closed curves). Use the solution formulas to determine the period of the trajectories.

Section 4. Matrix Methods

The **LinearAlgebra** package is introduced. Matrix and vector manipulation exercises provide practice via eigenvalue and eigenvector calculations. The matrix exponential is a key tool for solving constant coefficient linear systems of differential equations.

Use Maple to do the following. Start by loading the **LinearAlgebra** package using

```
with(LinearAlgebra):
```

1. Enter the matrix

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

using the three methods described below.

a. Using Matrix: `Matrix( [ [1,2,3], [4,5,6], [7,8,9] ] )`

b. In terms of its columns using the **< >** notation: `<<1,4,7>|<2,5,8>|<3,6,9>>`

c. In terms of its rows using the < > notation: `<<1|2|3>,<4|5|6>,<7|8|9>>`

2. Continuing 1. Enter the matrix in Exercise 1 in terms of its columns using the < > notation. Name it A.

a. Find the determinant and the characteristic polynomial of A.

b. Enter the vector $v = \langle 2, 3, 5 \rangle$ (use the entry `v := <2,3,5>`). It will appear as a column vector. Calculate the vector w = Av using the entry `w := A.v`. It will also be a column vector.

c. Calculate the dot product of v and w using `v.w`. (Both are column vectors, the product will be a scalar.) Verify that the entry `w.v` yields the same scalar as does `Transpose(v).Transpose(w)` and `Transpose(w).Transpose(v)`.

d. Calculate the product `Transpose(v).w`. This will be the same scalar as in part c conforming to the usual convention for such products. Now enter `v.Transpose(w)`. The output is a 3 x 3 rank 1 matrix (verify by entering `Rank(%)`).

3.* Define the three column vectors b1, b2, b3 as follows

```
b1,b2,b3 := <0,1,-1>,<1,1,0>,<-1,0,1>;
```

a. Define the matrix B having *b1, b2, b3* as its columns with the entry `B := <b1|b2|,`

b3>.

b. Find the characteristic polynomial of B and factor it with **`factor(%)`**.

c. Find B's eigenvalues with **`Eigenvalues(B)`**.

d. Find B's eigenvectors using **`Eigenvectors(B)`**. Name the input *lambda, V*.

e. Calculate **`MatrixInverse(V).B.V`**. What happens, and why?

4.* Continuing 3. Obtain the solution to v' = Bv satisfying $v(0) = \langle 1, 2, 3 \rangle$,

a. Do if first using **dsolve** applied to the system of linear differential equations defined by v' = Bv with the appropriate initial conditions.

b. Do it second by making a fundamental matrix solution $X(t)$ defined as the matrix with the eigenvector solutions in the columns then computing

$$v(t) = X(t)\, X(0)^{-1}\, v(0)$$

c. Do it third by using the matrix exponential, *Exp_At*. Once you have it, the solution is

$$v(t) = Exp\_At\; v(0)$$

d. Check the solution in each case.

Section 5. The Laplace Transform

The **inttrans** package contains procedures for calculating the Laplace transform and the inverse Laplace transform. Piecewise defined functions can be defined using the Heaviside function (unit step function in some texts). Dirac delta functions in the driver can be handled via Laplace transforms.

Use Maple to do the following problems. Load the inttrans package first via **`with(inttrans):`**. Define aliases for the Heaviside function and the Dirac delta using

`alias( H=Heaviside, delta=Dirac )`.

1. Obtain the Laplace transform of the following functions.

$$t\cos(3\,t)\ ,\ e^{-t}\sin(2\,t),\ \ \mathrm{H}(t - \mathrm{Pi})\ (t + \cos(t))\ \ ,\ sin(t) + \delta(t)\ \ ,$$

$$\text{piecewise}(\ t < 3, 0, t < 5, t - 3\ , 0\ )$$

2. Obtain the inverse Laplace transform of the following functions.

$$\frac{s}{s^2-4}\ ,\ \frac{s\,\mathrm{e}^{-2s}}{s^2-4}\ ,\ \frac{1}{s^2+4\,s-5}\ ,\ \frac{1}{s^2+4\,s-4}\ ,\ \frac{\mathrm{e}^{-3s}}{s^4}$$

3. Use the **convert** procedure to convert the piecewise function in Exercise 1 to Heaviside form and obtain the Laplace transform.

Note. Assuming that you have defined the aliases as indicated above, name the piecewise function Joe, and enter

`Jose := convert(Joe,H)`.

Plot *Joe* and *Jose* separately to check that they are the same. Compare the Laplace transforms of *Joe* and *Jose*. Are they also the same?

4. Consider the following initial value problem

$$y'' + y' + 9\,y = H(t - 2\,\mathrm{Pi}) \;\;,\;\; y(0) = 0\;, y'(0) = 1$$

a. Obtain the solution using **dsolve**. Plot it for `t = 0..40` and explain the behavior of the solution curve.

b. Obtain the solution using **dsolve/method=laplace**. How does the solution formula compare to the formula obtained in part a?

5. Consider the following initial value problem

$$y'' + y' + 9\,y = H(t - 2\,\mathrm{Pi}) - H(t - 5\,\mathrm{Pi}) \;\;,\;\; y(0) = 0\;, y'(0) = 1$$

a. Obtain the solution using **dsolve**. Plot it for `t = 0..40` and explain the behavior of the solution curve.

b. Obtain the solution using **dsolve/method=laplace**. How does the solution formula compare to the formula obtained in part a?

6.* Consider the following initial value problem

$$y'' + y' + 9\,y = H(t - 2\,\mathrm{Pi}) - H(t - 5\,\mathrm{Pi}) + 2\,\mathrm{delta}(t - 9\,\mathrm{Pi}) \;\;,\;\; y(0) = 0\;, y'(0) = 1$$

a. Obtain the solution using **dsolve**. Plot it for `t = 0..40` and explain the behavior of the solution curve.

b. Obtain the solution using **dsolve/method=laplace**. How does the solution formula compare to the formula obtained in part a?

7.* Use **unapply** to convert the solution to 6 a into a function, g. Use g to plot the phase plane trajectory and as well as the state space trajectory for the system. Use the following options in the spacecurve procedure for the state space trajectory

```
axes=boxed, orientation=[-60,70], color=red,
numpoints=400, labels=["Position","Velocity","Time"]
```

8.* Continuing 7. Use the entry

```
plots[animate]( plot, [ [g(t),D(g)(t),t=0..T] ],
                T=0..40, frames=82);
```

to plot an animation of the phase space trajectory. Click on the plot and use the video controls in the context bar to play the animation and also to view it one frame at a time. Compare the trajectory movement to the time series for position.

Note. The slide control can also be used to control the display of the frames. Assuming t is measured in seconds, you can reduce the frame rate to 2 frames per second to view the trajectory in "real time".